/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */
package org.grails.gradle.plugin.core

import groovy.transform.PackageScope

import org.gradle.api.Project
import org.gradle.api.artifacts.ProjectDependency
import org.gradle.api.artifacts.dsl.DependencyHandler
import org.gradle.api.attributes.AttributeContainer

import org.grails.gradle.plugin.exploded.GrailsExplodedPlugin

/**
 * Tracks Grails Plugins & handles prioritizing using the exploded variants for easier class reloading during development
 */
@PackageScope
class PluginDefiner implements DependencyHandler, GroovyInterceptable {

    @Delegate
    DependencyHandler target  // IDE/types; not used for the actual calls due to interception

    private final Project project
    boolean grailsRun

    List<ProjectDependency> dependencies = []

    PluginDefiner(Project project) {
        this.project = project
        this.target = project.dependencies
    }

    def invokeMethod(String name, Object objArgs) {
        def argArray = (objArgs instanceof Object[]) ? objArgs : [objArgs] as Object[]

        def methodMethod = target.metaClass.getMetaMethod(name, argArray)
        def result = (methodMethod ? methodMethod.invoke(target, argArray) : target.invokeMethod(name, argArray))

        if (result instanceof ProjectDependency) {
            ProjectDependency dependency = (ProjectDependency) result
            if (grailsRun) {
                dependency.attributes { AttributeContainer ac ->
                    ac.attribute(GrailsExplodedPlugin.EXPLODED_ATTRIBUTE, true)
                }
            }
            dependencies << dependency
        }

        return result
    }
}
